```
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

                     00001             LIST       N=83,C=110,B=4,P=16F18313
                     00002             ERRORLEVEL  -302   ;ignore banking messages
                     00003     title    "----- Fiddle Yard Track Control Program, (C) LWH brainware 2021"
                     00004     subtitle "----- Main Program"
                     00005 ;=========================================================================
                     00006 ;-   Fiddle Yard Track Control Program
                     00007 ;-   (C) LWH brainware 2021
                     00008 ;=========================================================================
                     00009 ;-   21-04-29     lwh created
                     00010 ;-   22-06-21     lwh info output implemented for Version 2
                     00011 ;-------------------------------------------------------------------------
                     00012     #include    "P16F18313.inc"
                     00001         LIST
                     00002
                     00003 ;==========================================================================
                     00004 ; Build date : May 15 2016
                     00005 ;  MPASM PIC16F18313 processor include
                     00006 ;
                     00007 ;  (c) Copyright 1999-2016 Microchip Technology, All rights reserved
                     00008 ;==========================================================================
                     00009
                     02267         LIST
                     00013 #define TESTSIM
                     00014 ;-------------------------------------------------------------------------
                     00015 ;-   General Adresses and Registers for PIC16F182x
                     00016 ;-------------------------------------------------------------------------
                     00017 ;-   Addresses
                     00018 ;-------------------------------------------------------------------------
  00000100           00019 CODE_START      EQU     0x100       ; start of program space
  00000000           00020 RESET_ADR       EQU     0x000       ; reset vector
  00000004           00021 INTER_ADR       EQU     0x004       ; interrupt vector
                     00022         ; last common RAM address declared so far (0070..007F)
  00000070           00023 LASTCOMMON      SET     0x70        ; initialize to 0070, updated later
                     00024 ;-------------------------------------------------------------------------
                     00025 ;-       Memory registers
                     00026 ;-------------------------------------------------------------------------
  00000070           00027 TRACKSTATUS     EQU LASTCOMMON      ; status bits
  00000000           00028 WAITLS          EQU     0           ; bit 0: waiting for LS dark
  00000001           00029 WAITFREE        EQU     1           ; bit 1: waiting for track free
                     00030
  00000071           00031 LASTCOMMON      SET TRACKSTATUS + 1 ; update next available RAM address
                     00032 #if  (LASTCOMMON > H'7F')
                     00033     error "Running out of common memory!"
                     00034 #endif
                     00035 ;-------------------------------------------------------------------------
                     00036 ;-  User ID words 0..3
                     00037 ;-
                     00038 UserID  code    0x8000
  8000   0103        00039         dw      0x0103      ; FY Track Control
  8001   3FFF        00040         dw      0x3FFF      ; not used
  8002   0000        00041         dw      0x0000      ; board version 0.0
  8003   0210        00042         dw      0x0210      ; version 2.1
                     00043 ;-------------------------------------------------------------------------
                     00044 ;-  Configuration bits
                     00045 ;-   - ecternal oscillator off
                     00046 ;-   - use internal oscillator with 1MHz
                     00047 ;-   - no clock output on pin OSC2
                     00048 ;-   - clock switching by software allowed
                     00049 ;-   - Fail-Safe Clock Monitor is disabled
  8007   1FEC        00050     __config    _CONFIG1, _FEXTOSC_OFF & _RSTOSC_HFINT1 & _CLKOUTEN_OFF & _CSWEN_O
                           N & _FCMEN_OFF & H'3FFF'
                     00051 ;-   - debugging off unless managed by debugger tool
                     00052 ;-   - Stack Overflow or Underflow will cause a Reset
                     00053 ;-   - PPS can be used repeatedly
                     00054 ;-   - Brown-out reset voltage is low (2.45V)
                     00055 ;-   - Brown-out will cause a reset
                     00056 ;-   - watchdog timer is off
                     00057 ;-   - power-up timer is on
                     00058 ;-   - RA3 is not !MCLR
  8008   37F0        00059     __config    _CONFIG2, _DEBUG_OFF & _STVREN_ON & _PPS1WAY_OFF & _BORV_LOW & _BO
                           REN_ON & _WDTE_OFF & _PWRTE_ON & _MCLRE_OFF & H'3FFF'
                     00060 ;-   - low voltage programming off
                     00061 ;-   - write-protection off (for now)
  8009   1FFF        00062     __config    _CONFIG3, _LVP_OFF & _WRT_OFF & H'3FFF'
                     00063 ;-   - EEPROM code protection off
                     00064 ;-   - Program code protection off (for now)
  800A   3FFF        00065     __config    _CONFIG4, _CPD_OFF & _CP_OFF & H'3FFF'
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
----- Main Program
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

                     00066
                     00067     ORG RESET_ADR                 ; reset vector
                     00068 ;--------------------------------------------------------------------------
                     00069 ;-  Program
                     00070 ;--------------------------------------------------------------------------
0000   2???          00071         GOTO    init              ; start with initialization
                     00072
                     00073     ORG INTER_ADR                 ; interrupt vector
                     00074 ;--------------------------------------------------------------------------
                     00075 ;-  Interrupt handling
                     00076 ;--------------------------------------------------------------------------
0004                 00077 intINT:     ; check general interrupt
0004   0020          00078         banksel PIR0
0005   1010          00079         BCF     PIR0, INTF        ; clear interrupt (not used here)
                     00080         ;-----------------------------------------------------------
0006                 00081 intButton:  ; check input state changed
0006   0020          00082         banksel PIR0
0007   1E10          00083         BTFSS   PIR0, IOCIF       ; check if interrupt on change is set
0008   2???          00084         GOTO    intTmr0           ; if not, check next
                     00085         ; handle input change interrupts
0009   1210          00086         BCF     PIR0, IOCIF       ; first, clear interrupt flag
000A   2???          00087         CALL    onButtonDown      ; then act on buttons, if any
000B   2???          00088         CALL    onTrackFree       ;
                     00089         ;-----------------------------------------------------------
000C                 00090 intTmr0:    ; check timer 0 interrupt (button debouncing counter)
000C   0020          00091         banksel PIR0
000D   1E90          00092         BTFSS   PIR0, TMR0IF      ; check if timer0 interrupt is set
000E   2???          00093         GOTO    intLS             ; if not, check next
                     00094         ; handle timer0 interrupts
000F   1290          00095         BCF     PIR0, TMR0IF      ; first, clear interrupt flag
0010   2???          00096         CALL    onButtonPoll
                     00097         ;-----------------------------------------------------------
0011                 00098 intLS:      ; check light barrier interrupt
0011   0020          00099         banksel PIR2
0012   1E92          00100         BTFSS   PIR2,   C1IF      ; check if comparator interrupt is set
0013   2???          00101         GOTO    intTmr2
                     00102         ; handle LS interrupts
0014   1292          00103         BCF     PIR2,   C1IF      ; first, clear interrupt flag
0015   2???          00104         CALL    onLSDark          ; then act on interrupt
                     00105         ;-----------------------------------------------------------
0016                 00106 intTmr2:    ; check timer 2 interrupt (info output update)
0016   0020          00107         banksel PIR1
0017   1C91          00108         BTFSS   PIR1, TMR2IF      ; check if timer2 interrupt is set
0018   2???          00109         GOTO    intDone           ; if not, check next
                     00110         ; handle timer2 interrupts
0019   1091          00111         BCF     PIR1, TMR2IF      ; first, clear interrupt flag
001A   2???          00112         CALL    onInfoUpdate
                     00113         ;-----------------------------------------------------------
001B                 00114 intDone:    ; all interrupts checked and handled
001B   0009          00115         RETFIE                    ; return from interrupt
                     00116
                     00117     ORG CODE_START
                     00118 ;--------------------------------------------------------------------------
                     00119 ;-  Controller Initialization
                     00120 ;--------------------------------------------------------------------------
                     00121 ;-  internal oscillator setup to 4MHz
0100                 00122 init:
0100   0032          00123         banksel OSCCON1
0101   3060          00124         MOVLW   B'01100000'
                     00125         ;          x|||++++---   postscaler set to 1:1
                     00126         ;           +++-------   HF internal oscillator, 1MHz (just in case)
0102   0099          00127         MOVWF   OSCCON1
0103   3003          00128         MOVLW   B'00000011'
                     00129         ;          xxxx++++---   HFINTOSC set to 4MHZ
0104   009F          00130         MOVWF   OSCFRQ
0105   3000          00131         MOVLW   H'00'
0106   009E          00132         MOVWF   OSCTUNE             ; fix fix osctune bug (from the errata doc)
0107   171D          00133         BSF     OSCEN,  HFOEN       ; enable oscillator
                     00134         ; now there may be a 2µs delay until oscillator has changed
                     00135         ; wait until oscillator is ready
                     00136 #ifndef  TESTSIM
                     00137         BTFSS   OSCSTAT1, HFOR      ; is HFINTOSC ready?
                     00138         GOTO    $-1                 ; if not, test again
                     00139 #endif
0108   0021          00140         banksel PIE0
0109   1010          00141         BCF     PIE0, INTE          ; disable INT pin
                     00142 ;-  set DAC to supply reference voltage for comparators positive input
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
----- Main Program
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

010A   0022           00143         banksel DACCON0
010B   3080           00144         MOVLW   B'10000000'
                      00145         ;         |x|x||x+---   negative source is Vss (GND)
                      00146         ;         | | ++-----   positive source is Vdd (+5V)
                      00147         ;         | +--------   no DAC1OUT pin
                      00148         ;         +----------   DAC is enabled
010C   0098           00149         MOVWF   DACCON0
010D   3014           00150         MOVLW   D'20'           ;set DAC output to xx/32 * 5V
010E   0099           00151         MOVWF   DACCON1
                      00152
010F   2???           00153         CALL    setupLS         ; set up the light barrier input
0110   2???           00154         CALL    setupTrack      ; set up the track ios
0111   2???           00155         CALL    setupButtons    ; set up the button input(s)
0112   2???           00156         CALL    setupInfo       ; set up the info output
                      00157 ;-  lock peripheral setup (?)
0113   0020           00158         banksel INTCON
0114   138B           00159         BCF     INTCON, GIE         ; disable interrupts
0115   003C           00160         banksel PPSLOCK
                      00161         ; required sequence:
0116   3055           00162         MOVLW   0x55
0117   008F           00163         MOVWF   PPSLOCK
0118   30AA           00164         MOVLW   0xAA
0119   008F           00165         MOVWF   PPSLOCK
011A   140F           00166         BSF     PPSLOCK, PPSLOCKED  ; disable further writes
                      00167 ;-  enable interrupts
011B   0020           00168         banksel INTCON
011C   178B           00169         BSF     INTCON, GIE         ; reenable interrupts
011D   170B           00170         BSF     INTCON, PEIE        ; enable peripheral interrupts
                      00171 #ifdef  TESTSIM
                      00172 ;       call    eepromTest
                      00173 #endif
                      00174 ;--------------------------------------------------------------------------
011E   0000           00175 idle:   NOP
                      00176 ;   waiting for interrupts
011F   2???           00177         GOTO    idle
                      00178 ;--------------------------------------------------------------------------
                      00179 ;-  End main program
                      00180 ;--------------------------------------------------------------------------
                      00181     include "inc/FYT_LS.inc"
                      00001     subtitle    "-----  LS Section"
```

```
-----  Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----   LS Section
LOC   OBJECT CODE     LINE  SOURCE  TEXT
  VALUE

                      00002     PAGE
                      00003 ;=================================================================================
                      00004 ;-       Light Barrier include file
                      00005 ;-       (C) LWH brainware 2021
                      00006 ;=================================================================================
                      00007 ;-       21-04-29    lwh created
                      00008 ;---------------------------------------------------------------------------------
                      00009 ;-   Light barrier threshold uses the comparator
                      00010 ;-   Interupt is fired when input goes higher than reference voltage (dark)
                      00011 ;---------------------------------------------------------------------------------
                      00012 ;-   memory registers
                      00013 ;---------------------------------------------------------------------------------
                      00014 ;-   Common RAM
                      00015
                      00016 #if (LASTCOMMON > H'7F')
                      00017     error "Running out of common memory!"
                      00018 #endif
                      00019 ;---------------------------------------------------------------------------------
                      00020 ;-   Constants
                      00021 ; bits correspond to RA pins
  00000001            00022 LSIN            EQU .1              ; RA1 as analog input for receiver
                      00023 ;---------------------------------------------------------------------------------
                      00024 ;-   Configure/Remap ports:
                      00025 ;-   RA1 is LS/comparator input
  0120                00026 setupLS:
                      00027         ;   set RA1 as analog input
  0120   0020         00028         banksel PORTA
  0121   108C         00029         BCF     PORTA,  LSIN            ; clear port pin
  0122   0021         00030         banksel TRISA
  0123   148C         00031         BSF     TRISA,  LSIN            ; set to 1 (in)
  0124   0023         00032         banksel ANSELA
  0125   148C         00033         BSF     ANSELA, LSIN            ; set as analog input
                      00034         ;   setup comparator mode
  0126   0022         00035         banksel CM1CON0
  0127   3096         00036         MOVLW   B'10010110'
                      00037         ;         |Rx|x||+---   C1SYNC  no sync with timer1
                      00038         ;         |  | |+----   C1HYS   hysteresis enabled
                      00039         ;         |  | +-----   C1SP    speed mode (always 1)
                      00040         ;         |  +-------   C1POL   output inverted
                      00041         ;         +----------   C1ON    comparator on
  0128   0091         00042         MOVWF   CM1CON0
                      00043         ;   setup comparator pins
  0129   30A8         00044         MOVLW   B'10101000'
                      00045         ;         |||||+++---   C1NCH   connect negative to C1IN0-
                      00046         ;         ||+++------   C1PCH   connect positive to DAC (setup in main)
                      00047         ;         |+---------   C1INTN  no interrupt on negative transition
                      00048         ;         +----------   C1INTP  interrupt on positive transition (in > DAC
                           )
  012A   0092         00049         MOVWF   CM1CON1
                      00050         ;   interrupt on
  012B   0021         00051         banksel PIE2
  012C   1692         00052         BSF     PIE2,   C1IE    ; enable comparator interrupt
  012D   0008         00053     RETURN
                      00054 ;---------------------------------------------------------------------------------
                      00055 ;-  Return 1 in W when LS is dark, 0 when uncovered
  012E                00056 getLS:
  012E   0022         00057         banksel CM1CON0
  012F   1B11         00058         BTFSC   CM1CON0, C1OUT  ; check comparator output
  0130   3401         00059     RETLW       1               ; if dark (in > DAC), return with 1
  0131   3400         00060     RETLW       0               ; else return with 0
                      00061 ;---------------------------------------------------------------------------------
                      00062 ;-  when LS turns dark, stop incoming train
  0132                00063 onLSDark:
  0132   2???         00064         CALL    getLS
  0133   0809         00065         MOVF    WREG, W         ; affect zero flag
  0134   1903         00066         BTFSC   STATUS, Z       ; check zero flag
  0135   0008         00067     RETURN                      ; if set (LS not dark), do nothing
                      00068         ;-----------------------------------------------------------
                      00069         ;-  if not set (LS dark), check wait for bit
  0136   1C70         00070         BTFSS   TRACKSTATUS, WAITLS
  0137   0008         00071     RETURN                      ; if not set, do nothing
                      00072         ;-----------------------------------------------------------
                      00073         ;-  if set (incoming train), stop it
  0138   1070         00074         BCF     TRACKSTATUS, WAITLS ; clear wait for bit
  0139   2???         00075         CALL    trackOff        ; switch track off
  013A   0008         00076     RETURN
                      00077 ;=================================================================================
```

----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
----- LS Section

```
LOC  OBJECT CODE     LINE  SOURCE  TEXT
  VALUE

                     00182     include "inc/FYT_Track.inc"
                     00001     subtitle    "-----   Track Section"
```

```
-----  Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----   Track  Section
LOC   OBJECT CODE     LINE  SOURCE  TEXT
  VALUE

                      00002     PAGE
                      00003 ;=============================================================================
                      00004 ;-       Track include file
                      00005 ;-       (C) LWH brainware 2021
                      00006 ;=============================================================================
                      00007 ;-       21-04-29    lwh created
                      00008 ;-----------------------------------------------------------------------------
                      00009 ;-   Track (relais) in on on H
                      00010 ;-   Track occupied is L
                      00011 ;-----------------------------------------------------------------------------
                      00012 ;-   memory registers
                      00013 ;-----------------------------------------------------------------------------
                      00014 ;-   Common RAM
                      00015 #if  (LASTCOMMON > H'7F')
                      00016     error "Running out of common memory!"
                      00017 #endif
                      00018 ;-----------------------------------------------------------------------------
                      00019 ;-   Constants
  00000002            00020 RELAISOUT   EQU     2    ; output RA2 (relais on = H)
  00000000            00021 TRACKIN     EQU     0    ; input RA0 (occupied = L)
                      00022 ;-----------------------------------------------------------------------------
                      00023 ;-   Configure/Remap track i/os
013B                  00024 setupTrack:
013B   0020           00025         banksel PORTA
013C   110C           00026         BCF     PORTA,  RELAISOUT   ; relais off
013D   0022           00027         banksel LATA
013E   110C           00028         BCF     LATA,   RELAISOUT   ; latch off
013F   0021           00029         banksel TRISA
0140   110C           00030         BCF     TRISA,  RELAISOUT   ; tris bit = 0 (output)
0141   140C           00031         BSF     TRISA,  TRACKIN     ; tris bit = 1 (input)
0142   0023           00032         banksel ANSELA
0143   110C           00033         BCF     ANSELA, RELAISOUT   ; make port readable
0144   100C           00034         BCF     ANSELA, TRACKIN     ; digital input
0145   0024           00035         banksel WPUA
0146   140C           00036         BSF     WPUA,   TRACKIN     ; pull-up for input
0147   0025           00037         banksel ODCONA
0148   110C           00038         BCF     ODCONA, RELAISOUT   ; no open drain for output
                      00039     ;   interrupt on track free
0149   0027           00040         banksel IOCAP
014A   1411           00041         BSF     IOCAP,  TRACKIN     ; enable interrupt on positive edge
014B   1012           00042         BCF     IOCAN,  TRACKIN     ; disable interrupt on negative edge
014C   0020           00043         banksel PIR0
014D   1210           00044         BCF     PIR0,   IOCIF       ; clear any lingering interrupts
014E   0021           00045         banksel PIE0
014F   1610           00046         BSF     PIE0,   IOCIE       ; enable interrupt on change
0150   0008           00047     RETURN
                      00048 ;-----------------------------------------------------------------------------
                      00049 ;-   when track just became free, release track
0151                  00050 onTrackFree:
                      00051         ; check and clear interrupt flag
0151   0027           00052         banksel IOCAF
0152   1C13           00053         BTFSS   IOCAF,  TRACKIN     ; interrupt flag set?
0153   0008           00054     RETURN                          ; if not, nothing to do here
0154   1013           00055         BCF     IOCAF,  TRACKIN     ; clear interrupt flag
0155   0020           00056         banksel PORTA
0156   1C0C           00057         BTFSS   PORTA,  TRACKIN     ; still free?
0157   0008           00058     RETURN                          ; if not, ignore glitch
                      00059         ;-------------------------------------------------------------
                      00060         ;-   if set (track free), check wait for bit
0158   1CF0           00061         BTFSS   TRACKSTATUS, WAITFREE
0159   0008           00062     RETURN                          ; if not set, do nothing
                      00063         ;-------------------------------------------------------------
                      00064         ;-   if set (outgoing train), stop it
015A   10F0           00065         BCF     TRACKSTATUS, WAITFREE   ; clear wait for bit
015B   2???           00066         CALL    trackOff            ; switch track off
015C   0008           00067     RETURN
                      00068 ;-----------------------------------------------------------------------------
                      00069 ;-   switch track on
015D                  00070 trackOn:
015D   0020           00071         banksel PORTA               ; just in case
015E   150C           00072         BSF     PORTA,  RELAISOUT   ; set relais output 1
015F   0008           00073     RETURN
                      00074 ;-----------------------------------------------------------------------------
                      00075 ;-   switch track off
0160                  00076 trackOff:
0160   0020           00077         banksel PORTA               ; just in case
0161   110C           00078         BCF     PORTA,  RELAISOUT   ; set relais output 0
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----  Track Section
LOC  OBJECT CODE     LINE  SOURCE  TEXT
  VALUE

0162   0008           00079     RETURN
                      00080 ;-------------------------------------------------------------------------
                      00081 ;-  return 1 in w, if track occupied, else 0
0163                  00082 getOccupied:
0163   0020           00083         banksel PORTA               ; just in case
0164   180C           00084         BTFSC   PORTA, TRACKIN      ; check input
0165   3400           00085     RETLW       0                   ; if not 0, track is free
0166   3401           00086     RETLW       1                   ; if 0, track is occupied
                      00087 ;-------------------------------------------------------------------------
                      00088 ;-  return 1 in w, if track relais on, else 0
0167                  00089 getTrackOn:
0167   0020           00090         banksel PORTA
0168   190C           00091         BTFSC   PORTA, RELAISOUT    ; check output
0169   3401           00092     RETLW       1                   ; if on, return 1
016A   3400           00093     RETLW       0                   ; if off, return 0
                      00094 ;=========================================================================
                      00183     include "inc/FYT_Buttons.inc"
                      00001     subtitle    "-----   Button Section"
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----  Button Section
LOC  OBJECT CODE     LINE  SOURCE  TEXT
  VALUE

                     00002     PAGE
                     00003 ;=============================================================================
                     00004 ;-       Button include file
                     00005 ;-       (C) LWH brainware 2021
                     00006 ;=============================================================================
                     00007 ;-       21-04-29     lwh created
                     00008 ;-       22-06-19     lwh release button removed, not used
                     00009 ;-----------------------------------------------------------------------------
                     00010 ;-   Button inputs are active low with internal pullup, debounced by timer
                     00011 ;-----------------------------------------------------------------------------
                     00012 ;-   memory registers
                     00013 ;-----------------------------------------------------------------------------
                     00014 ;-   Common RAM
  00000071           00015 LASTBTNS        EQU LASTCOMMON      ; last button state
  00000072           00016 LASTCOMMON      SET LASTBTNS + 1    ; update next available RAM address
                     00017 #if  (LASTCOMMON > H'7F')
                     00018     error "Running out of common memory!"
                     00019 #endif
                     00020 ;-----------------------------------------------------------------------------
                     00021 ;-   Constants
  0000009B           00022 TMR0PERIOD      EQU .155            ; for a timer period of approx 20ms
                     00023 ; bits correspond to RA pins
  00000004           00024 TRACKBN         EQU .4              ; bit 4, L for track button pressed
                     00025 ;-----------------------------------------------------------------------------
                     00026 ;-   Configure/Remap button inputs:
                     00027 ;-   RA4 is track button input
  016B               00028 setupButtons:
  016B   0021        00029         banksel TRISA
  016C   160C        00030         BSF     TRISA,  TRACKBN     ; trackbn pin is 1 (input)
  016D   0024        00031         banksel WPUA
  016E   160C        00032         BSF     WPUA,   TRACKBN     ; enable weak pullup
  016F   0023        00033         banksel ANSELA
  0170   120C        00034         BCF     ANSELA, TRACKBN     ; make sure its digital input
                     00035         ; set timer0 to poll buttons every 20 ms
  0171   0020        00036         banksel T0CON0
  0172   3000        00037         MOVLW   B'00000000'
                     00038         ;         |xr|++++---   postscaler 1:1
                     00039         ;         |  +-------   8 bit timer only
                     00040         ;         +----------   timer disabled for now
  0173   0097        00041         MOVWF   T0CON0
  0174   0020        00042         banksel T0CON1
  0175   3057        00043         MOVLW   B'01010111'
                     00044         ;         ||||++++---   prescaler 1:128 for period * 128us
                     00045         ;         |||+-------   no synchronization (no external clock)
                     00046         ;         +++--------   clock source command clock (Fosc/4)
  0176   0098        00047         MOVWF   T0CON1
  0177   0020        00048         banksel TMR0H
  0178   309B        00049         MOVLW   TMR0PERIOD
  0179   0096        00050         MOVWF   TMR0H               ; set period
  017A   2???        00051         CALL    enableBnDnInt       ; prepare for interrupts on button down
  017B               00052 enableBnInt:
  017B   0020        00053         banksel PIR0
  017C   1290        00054         BCF     PIR0, TMR0IF        ; clear any lingering interrupts
  017D   0021        00055         banksel PIE0
  017E   1690        00056         BSF     PIE0, TMR0IE
  017F   0008        00057     RETURN
                     00058 ;-----------------------------------------------------------------------------
                     00059 ;-   enable interrupt on button down (negative flank)
                     00060 ;-   make sure to enable peripheral interrupts in main program
  0180               00061 enableBnDnInt:
  0180   0027        00062         banksel IOCAP
  0181   1612        00063         BSF     IOCAN, TRACKBN      ; set interrupt on negative flank
  0182   1211        00064         BCF     IOCAP, TRACKBN      ; clear interrupt on positive flank
  0183   0021        00065         banksel PIE0
  0184   1610        00066         BSF     PIE0, IOCIE         ; enable interrupt on change
  0185   0008        00067     RETURN
                     00068 ;-----------------------------------------------------------------------------
                     00069 ;-   enable interrupt on button up (positive flank)
                     00070 ;-   make sure to enable peripheral interrupts in main program
  0186               00071 enableBnUpInt:
  0186   0027        00072         banksel IOCAP
  0187   1611        00073         BSF     IOCAP, TRACKBN      ; set interrupt on positive flank
  0188   1212        00074         BCF     IOCAN, TRACKBN      ; clear interrupt on negative flank
  0189   0021        00075         banksel PIE0
  018A   1610        00076         BSF     PIE0, IOCIE         ; enable interrupt on change
  018B   0008        00077     RETURN
                     00078 ;-----------------------------------------------------------------------------
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----  Button Section
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

                     00079 ;-  when button pressed, start timer and wait for timer interrupt
018C                 00080 onButtonDown:
                     00081         ;   check and clear interrupt flags
018C   0027           00082         banksel IOCAF
018D   1A13           00083         BTFSC   IOCAF, TRACKBN
018E   1671           00084         BSF     LASTBTNS, TRACKBN   ; set track button flag, if RA4 interrupt
018F   1213           00085         BCF     IOCAF, TRACKBN      ; clear interrupt
0190   08F1           00086         MOVF    LASTBTNS, F         ; check whether any button flag set
0191   1903           00087         BTFSC   STATUS, Z           ; is zero flag set?
0192   0008           00088     RETURN                          ; if yes, no button, nothing to do here
                     00089         ;   a button flag is set, so start timer
0193   0020           00090         banksel TMR0H
0194   309B           00091         MOVLW   TMR0PERIOD
0195   0096           00092         MOVWF   TMR0H               ; set timer period
0196   0020           00093         banksel T0CON0
0197   1797           00094         BSF     T0CON0, T0EN        ; (re)start timer
0198   0008           00095     RETURN
                     00096 ;--------------------------------------------------------------------------
                     00097 ;-  when button released, start timer and wait for timer interrupt
0199                 00098 onButtonUp:
                     00099         ;   check and clear interrupt flags
0199   0027           00100         banksel IOCAF
019A   1A13           00101         BTFSC   IOCAF, TRACKBN
019B   1671           00102         BSF     LASTBTNS, TRACKBN   ; set track button flag, if RA4 interrupt
019C   1213           00103         BCF     IOCAF, TRACKBN      ; clear interrupt
019D   08F1           00104         MOVF    LASTBTNS, F         ; check whether any button flag set
019E   1903           00105         BTFSC   STATUS, Z           ; is zero flag set?
019F   0008           00106     RETURN                          ; if yes, no button, nothing to do here
                     00107         ;   a button flag is set, so start timer
01A0   0020           00108         banksel TMR0H
01A1   309B           00109         MOVLW   TMR0PERIOD
01A2   0096           00110         MOVWF   TMR0H               ; set timer period
01A3   0020           00111         banksel T0CON0
01A4   1797           00112         BSF     T0CON0, T0EN        ; (re)start timer
01A5   0008           00113     RETURN
                     00114 ;--------------------------------------------------------------------------
                     00115 ;-  react on button timer interrupt, which occurs first 20ms after downward
                     00116 ;-  flank, then each 20ms
                     00117 ;-  check if button is still down, if yes, act accordingly
01A6                 00118 onButtonPoll:
01A6   1A71           00119         BTFSC   LASTBTNS, TRACKBN   ; was track button pressed?
01A7   2???           00120         CALL    checkTrackBn        ; if yes, check it now
                     00121         ; all buttons cared for
01A8   0020           00122         banksel T0CON0
01A9   1397           00123         BCF     T0CON0, T0EN        ; no more buttons, stop timer
01AA   0008           00124     RETURN
01AB                 00125 checkTrackBn:
01AB   0020           00126         banksel PORTA
01AC   1E0C           00127         BTFSS   PORTA, TRACKBN      ; check track button input
01AD   2???           00128         GOTO    TrackBnDown         ; if (still) L (pressed), act on it
01AE   1271           00129         BCF     LASTBTNS, TRACKBN   ; if not, clear track button flag
01AF   0008           00130     RETURN
                     00131 ;--------------------------------------------------------------------------
                     00132 ;-  react on track button pressed
                     00133 ;-  if track occupied, set for departure, wait for track free
                     00134 ;-  if track free, set for arrival, wait for LS dark
01B0                 00135 TrackBnDown:
01B0   1271           00136         BCF     LASTBTNS, TRACKBN   ; clear button flag, we recognized it.
01B1   2???           00137         CALL    trackOn             ; switch the track on
01B2   2???           00138         CALL    getOccupied
01B3   1809           00139         BTFSC   WREG, 0             ; check whether its 1 (occupied)
01B4   2???           00140         GOTO    onOutgoing          ; if 1, jump
                     00141         ;   else track is free, train incoming
01B5                 00142 onIncoming:
01B5   1470           00143         BSF     TRACKSTATUS, WAITLS ; wait for light barrier dark
                     00144         ; enable interrupt?
01B6   0008           00145     RETURN
01B7                 00146 onOutgoing:
01B7   14F0           00147         BSF     TRACKSTATUS, WAITFREE; wait for track free
                     00148         ; enable interrupt?
01B8   0008           00149     RETURN
                     00150 ;==========================================================================
                     00184     include "inc/FYT_Info.inc"
                     00001     subtitle    "-----  Info Section"
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----  Info Section
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

                     00002     PAGE
                     00003 ;*****************************************************************************
                     00004 ;-       Info include file
                     00005 ;-       (C) LWH brainware 2022
                     00006 ;*****************************************************************************
                     00007 ;-       22-06-20    lwh created
                     00008 ;-----------------------------------------------------------------------------
                     00009 ;-   Outputs info on one line: Occupied (L), Selected (H), nothing (open).
                     00010 ;-   States are multiplexed over time.
                     00011 ;-   Uses timer2 with interrupt for multiplexing
                     00012 ;-----------------------------------------------------------------------------
                     00013 ;-   memory registers
                     00014 ;-----------------------------------------------------------------------------
                     00015 ;-   Common RAM
  00000072           00016 MUXPHASE        EQU LASTCOMMON      ; current multiplex phase (0 or 1)
  00000073           00017 INFOBITS        EQU MUXPHASE + 1    ; bits for current info, 0 if no info
  00000074           00018 LASTCOMMON      SET INFOBITS + 1    ; update next available RAM address
                     00019 #if (LASTCOMMON > H'7F')
                     00020     error "Running out of common memory!"
                     00021 #endif
                     00022 ;-----------------------------------------------------------------------------
                     00023 ;-   Constants
  0000003C           00024 TMR2PERIODF     EQU .60             ; for a timer period of approx 1ms
  000000FA           00025 TMR2PERIODS     EQU .250            ; for a timer period of approx 250ms
                     00026 ; bits correspond to RA pins
  00000005           00027 INFOPIN         EQU .5              ; PIN RA5
                     00028 ; info flag pins
  00000000           00029 IBOCC           EQU .0              ; H if occupied
  00000001           00030 IBTRK           EQU .1              ; H if relais on
                     00031 ;-----------------------------------------------------------------------------
                     00032 ;-   Configure/Remap button inputs:
                     00033 ;-   RA4 is track button input
01B9                 00034 setupInfo:
01B9   0021          00035         banksel TRISA
01BA   168C          00036         BSF     TRISA,  INFOPIN     ; info pin is 1 (input) for open state
01BB   0024          00037         banksel WPUA
01BC   128C          00038         BCF     WPUA,   INFOPIN     ; disable weak pullup
01BD   0023          00039         banksel ANSELA
01BE   128C          00040         BCF     ANSELA, INFOPIN     ; make sure its digital IO
                     00041         ; set timer2 for continuous running
01BF   0020          00042         banksel T2CON
                     00043     ;   MOVLW   B'00000010'          ; setup for 1000Hz
                     00044         ;          x|||||++----- T2CKPS prescaler 1:16
                     00045         ;           ||||+------- TMR2ON off for now
                     00046         ;           ++++-------- T2OUTPS postscaler 1:1
01C0   307B          00047         MOVLW   B'01111011'         ; setup for 4Hz
                     00048         ;          x|||||++----- T2CKPS prescaler 1:64
                     00049         ;           ||||+------- TMR2ON off for now
                     00050         ;           ++++-------- T2OUTPS postscaler 1:16
01C1   009F          00051         MOVWF   T2CON
01C2   0020          00052         banksel PR2
01C3   30FA          00053         MOVLW   TMR2PERIODS
01C4   009E          00054         MOVWF   PR2                 ; set timer period
01C5   0020          00055         banksel T2CON
01C6   151F          00056         BSF     T2CON, TMR2ON       ; start timer2
01C7                 00057 enableInfoInt:
01C7   0020          00058         banksel PIR1
01C8   1091          00059         BCF     PIR1, TMR2IF        ; clear any lingering interrupts
01C9   0021          00060         banksel PIE1
01CA   1491          00061         BSF     PIE1, TMR2IE
01CB   0008          00062     RETURN
                     00063 ;-----------------------------------------------------------------------------
                     00064 ;-   act on timer2 clock signal
                     00065 ;-   if phase 0, output occupied state, if phase 1 selected state
01CC                 00066 onInfoUpdateOld:
01CC   0021          00067         banksel TRISA
01CD   168C          00068         BSF     TRISA, INFOPIN      ; switch info pin off
01CE   1872          00069         BTFSC   MUXPHASE, 0         ; check phase
01CF   2???          00070         GOTO    onPhaseSel          ; if 1, output selected state
                     00071         ; output occupied state, if so
01D0                 00072 onPhaseOcc:
01D0   1472          00073         BSF     MUXPHASE, 0         ; change phase for next call
01D1   2???          00074         CALL    getOccupied
01D2   1C09          00075         BTFSS   WREG,   0           ; check whether its 1 (occupied)
01D3   0008          00076     RETURN                          ; if not, nothing to do here
01D4   0022          00077         banksel LATA
01D5   128C          00078         BCF     LATA, INFOPIN       ; set low for occupied
```

```
----- Fiddle Yard Track Control Program, (C) LWH brainware 2021
-----  Info Section
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

01D6   0021           00079         banksel TRISA
01D7   128C           00080         BCF     TRISA, INFOPIN      ; switch info pin on
01D8   0008           00081     RETURN
                      00082         ; output selected state, if so
01D9                  00083 onPhaseSel:
01D9   1072           00084         BCF     MUXPHASE, 0         ; change phase for next call
01DA   2???           00085         CALL    getTrackOn
01DB   1C09           00086         BTFSS   WREG,   0           ; check whether its 1 (track selected)
01DC   0008           00087     RETURN                          ; if not, nothing to do here
01DD   0022           00088         banksel LATA
01DE   168C           00089         BSF     LATA, INFOPIN       ; set high for selected
01DF   0021           00090         banksel TRISA
01E0   128C           00091         BCF     TRISA, INFOPIN      ; switch info pin on
01E1   0008           00092     RETURN
                      00093 ;--------------------------------------------------------------------------
                      00094 ;-  act on timer2 clock signal
                      00095 ;-  if phase 0, output occupied state, if phase 1 selected state
01E2                  00096 onInfoUpdate:
01E2   09F2           00097         COMF    MUXPHASE, F         ; change phase byte
01E3   01F3           00098         CLRF    INFOBITS            ; clear info bits
01E4   2???           00099         CALL    getOccupied
01E5   1809           00100         BTFSC   WREG,   0           ; check whether its 1 (occupied)
01E6   1473           00101         BSF     INFOBITS, IBOCC     ; if yes, set occupied bit
01E7   2???           00102         CALL    getTrackOn
01E8   1809           00103         BTFSC   WREG,   0           ; check whether its 1 (track selected)
01E9   14F3           00104         BSF     INFOBITS, IBTRK     ; if yes, set track bit
01EA   08F3           00105         MOVF    INFOBITS, F         ; check if any info bit set (not zero)
01EB   1D03           00106         BTFSS   STATUS, Z           ;
01EC   2???           00107         GOTO    checkOccupiedOn     ; if yes, check which
                      00108         ; no info bit set, so switch output off
01ED   0021           00109         banksel TRISA
01EE   168C           00110         BSF     TRISA, INFOPIN      ; switch info pin off
01EF   0008           00111     RETURN
                      00112         ; check whether occupied needs to be signalled
01F0                  00113 checkOccupiedOn:
01F0   1C72           00114         BTFSS   MUXPHASE, 0         ; check whether its occupied phase
01F1   2???           00115         GOTO    checkTrackOn        ; if not, check track
                      00116         ; - - - - - - -
01F2   1C73           00117         BTFSS   INFOBITS, IBOCC     ; is occupied bit set?
01F3   0008           00118     RETURN                          ; if not, nothing to do
                      00119         ; signal occupied by setting output L
01F4   0022           00120         banksel LATA
01F5   128C           00121         BCF     LATA, INFOPIN       ; set low for occupied
01F6   0021           00122         banksel TRISA
01F7   128C           00123         BCF     TRISA, INFOPIN      ; switch info pin on
01F8   0008           00124     RETURN
                      00125         ; check whether track selected needs to be signalled
01F9                  00126 checkTrackOn:
01F9   1872           00127         BTFSC   MUXPHASE, 0         ; check whether its track phase
01FA   2???           00128         GOTO    checkOccupiedOn     ; if not, check occupied
                      00129         ; - - - - - - -
01FB   1CF3           00130         BTFSS   INFOBITS, IBTRK     ; is track bit set?
01FC   0008           00131     RETURN                          ; if not, nothing to do
                      00132         ; signal selected by setting output H
01FD   0022           00133         banksel LATA
01FE   168C           00134         BSF     LATA, INFOPIN       ; set high for selected
01FF   0021           00135         banksel TRISA
0200   128C           00136         BCF     TRISA, INFOPIN      ; switch info pin on
0201   0008           00137     RETURN
                      00185     subtitle    ""
                      00186 ;==========================================================================
1FEC 37F0 1FFF 3FFF   00187         END                         ; end of program code
```